8. The High Level Language[1]

"High thoughts need a high language."

(Aristophanes, 448-380 BC)

This chapter presents an overview and specification of the *Jack* programming language. Jack is a simple object-based language that can be used to write high-level programs. It has the basic features and flavor of modern object-oriented languages like Java, with a much simpler syntax and no support for inheritance. The introduction of Jack marks the beginning of the end of our journey. In chapters 9 and 10 we will write a compiler that translates Jack programs into VM code, and in Chapter 11 we will develop a simple operating system for the Jack/Hack platform, written in Jack. This will complete the computer's construction.

The chapter starts with a brief history of programming languages, followed by an overview and formal description of the Jack language. Next, we illustrate some computer games written in Jack, and give general guidelines on how to write similar interactive applications over the Hack platform. All theses programs can be compiled using a *Jack compiler* that we provide, and then run on the computer hardware that we built in chapters 1-5. Throughout the chapter, our goal is not to turn you into a Jack programmer. Instead, our "hidden agenda" is to prepare you to write the compiler and operating system described in the chapters that lie ahead. In that respect, Jack should be viewed mainly as a *tool*, designed to highlight the software engineering principles underlying the implementation of modern programming languages.

There exist many programming languages in the field, and most of them are either *procedural*, *functional*, or *object-oriented*. Every one of these programming paradigms has its strengths and limitations, and any comparison of them is clearly beyond the scope and purpose of this book. Yet one thing that we can safely say about *object-orientation* is that, for better or worse, it is the most complex programming paradigm among the three. Therefore, object-oriented languages are hard to master, and they require the most intricate compilers. With that in mind, understanding how object-oriented languages are designed, and how they work "under the hood", will give you expert knowledge of a complex and widely-used programming paradigm.

8.1 Background

First there were only machine languages, and programs were extremely hard to write and understand. Then came symbolic notation, and assembly languages followed. But the programs were still cumbersome and cryptic, and only highly skilled programmers could write and maintain them. Then in the mid 1950's, it was felt that symbolic computation could be taken a step further, allowing programs to contain expressions like `y=-b+sqrt(b^2-4*a*c)` instead of long series of machine instructions. Thus the Fortran language and the Fortran compiler were born. *Fortran*, for *FORmula TRANslation*, was an important breakthrough, as it placed programming in the hands of end-users. For the first time, scientists and engineers could write programs directly, without the assistance of professional programmers.

[1] From *The Elements of Computing Systems*, Nisan & Schocken, MIT Press, forthcoming in 2003, www.idc.ac.il/csd

But, not unlike assembly, Fortran was a rigid and clunky language. Laden with primitive subroutine calling and notorious for its "goto logic," Fortran programs resulted in "spaghetti code" that was difficult to read and maintain. The world of programming needed a paradigm shift, and this came from Europe in the mid 1960's in the form of Algol. *Algol*, for *ALGOrithmic Language*, was the first modern high level programming language. Equipped with structured control flow commands (eliminating the need for *goto*), dynamic memory management (allowing recursion) and an elegant syntax, Algol inspired and transformed the world of programming. In short order, many Algol-like languages followed, most notably *Pascal* and *C*.

But, not unlike Fortran, Pascal and C were procedural languages. It did not take long before people observed that a great deal of programming evolves not only around *processes*, but also around *data abstraction and representation*. Although languages like C and Pascal could handle versatile data structures, the processing of these structures was forced into a procedural (rather than data-centric) logic, resulting in complex and unnatural code. Once again, the next new thing came from Europe, this time in the form of object-orientation and C++. Following the wide spread use of C++ in the 1980's and the introduction of Java in the mid 1990's, object-oriented languages became the de-facto tools for industrial strength programming.

Compared to their humble ancestors, languages like C++ and Java are extremely complex, requiring sophisticated compilers and run-time environments. There exists a simpler version of the data-centric approach to programming, known as *object-based* languages. These languages (e.g. Visual Basic 6.0) can represent and manipulate objects, but they don't support inheritance and other advanced data abstraction mechanisms. In this chapter we introduce a simple object-based language, called *Jack*, whose look and feel resembles that of Java.

It's important to note at the outset that in and by itself, Jack is a rather uninteresting and simple-minded language. However, this simplicity has a purpose. First, you can learn (and unlearn) Jack very quickly -- in about an hour. Second, the Jack language was carefully planned to lend itself nicely to modern compilation techniques. As a result, one can write an elegant *Jack compiler* with relative ease, as we will do in Chapters 9 and 10. In other words, the deliberately simple structure of Jack will help us uncover the software engineering principles underlying modern languages like Java and C#. Rather than taking the compilers and run-time environments of these languages for granted, we will build a Jack compiler and a run-time environment ourselves, beginning in the next chapter. For now, let's take Jack out of the box.

8.2 The Jack Language

We begin with some illustrations of Jack programming, and continue with a formal language specification. The former is all that you need in order to start writing Jack programs, and the latter is a complete language reference.

8.2.1 Example

The task: Every programming language has a fixed set of primitive data types, of which Jack supports three: `int`, `char`, and `boolean`. Suppose we wish to endow Jack with the added ability to handle *rational numbers*, i.e. objects of the form n/m where n and m are integers. This can be

done by creating a stand-alone class, designed to provide a fraction abstraction for Jack programs. Let us call this class `Fraction`.

Defining the class interface: A reasonable way to get started is to specify the set of properties and services that one would normally expect to see in a fraction abstraction. One such *Application Program Interface*, or *API*, is given in Prog. 8-1.

Fraction (partial API):

// A "Fraction" is an object representation of *n*/*m* where *n* and *m* are integers (e.g. 17/253)

`field int numerator:` Numerator of *this* object.

`field int denominator:` Denominator of *this* object.

`constructor Fraction new(int a,int b):` Returns a new Fraction object.

`method int getNumerator():` Returns the numerator of *this* object.

`method int getDenominator():` Returns the denominator of *this* object.

`function Fraction add(Fraction f, Fraction g):` Returns the Fraction sum of `f` and `g`.

`method void print():` Prints *this* object in the format "numerator/denominator".

PROGRAM 8-1: Proposed API for the Fraction abstraction. In Jack, object properties are represented by *fields*, operations on the current object (referred to as *this* object) are represented by *methods*, and class-level operations are represented by *functions*.

Using the class (Example I): APIs mean different things to different people. If you are the programmer who has to *implement* the fraction abstraction, you can view its API as a contract that must be implemented in some way. Alternatively, if you are a programmer who needs to *use* fractions in your work, you can view the API as a library of fraction objects generators and fraction-related operations. Taking this latter view, consider the Jack code listed in Prog. 8-2.

```
// Compute the sum of 2/3 and 1/5
var Fraction a,b,c;   // declare some Fraction variables
let a=Fraction.new(2,3);  // create a=2/3
let b=Fraction.new(1,5);  // create b=1/5
let c=Fraction.add(a,b);  // compute c=a+b
do c.print();             // print c
// the above code should print the text: "13/15"
```

PROGRAM 8-2: Using the Fraction abstraction in Jack programs

Prog. 8-2 illustrates several features of the Jack language: declaration and creation of new objects, use of the assignment statement `let`, and method calling using the `do` statement. As the code implies, users of the fraction abstraction don't have to know anything about its underlying *implementation.* Rather, they should be given access only to the fraction *interface*, or API, and then use it as a black box server of fraction-related operations.

Implementing the class: We now turn to the other player in our story -- the programmer who has to actually implement the fraction API in Jack. One possible implementation is given in Prog. 8-3. This example illustrates several additional features of object-oriented programming in Jack: a typical program structure (*classes*, *methods*, *constructors*, *functions*), some statements (*let*, *do*, *return*), some variable types (*fields*, *locals*, *parameters*), and use of printing methods from the language's `Output` library.

```
/** A fraction (partial implementation) */

class Fraction {

  field int numerator, denominator;

  /** Construct a new fraction given a numerator and denominator */
  constructor Fraction new(int a, int b){
    let numerator=a;
    let denominator=b;
    return this;
  }

  method int getNumerator(){
    return numerator;
  }

  method int getDenominator(){
    return denominator;
  }

  /** Return a fraction that is the sum of two input fractions */
  // This implementation is somewhat buggy as it does not reduce
  // the fraction: (a/b)+(c/d)=(a*d+c*b)/(b*d)

  function Fraction add(Fraction f, Fraction g){
  var int sum;
    let sum = (f.getNumerator()*g.getDenominator())
              +(g.getNumerator()*f.getDenominator());
    return Fraction.new(sum, f.getDenominator()*g.getDenominator());
  }

  // print the fraction, using printing methods from the
  // language's Output library (yet another Jack class).
  method void print() {
    do Output.printInt(numerator);
    do Output.printString(" / ");
    do Output.printInt(denominator);
    return;
  }

} // end Fraction
```

PROGRAM 8-3: a Fraction class implementation, in Jack. Illustrates the use of *fields* (similar to Java's *private variables*), accessor methods, and a *function* (similar to Java's *static method*).

Using the class (Example II): We end this section with one more illustrative use of the fraction abstraction. Suppose we have to compute the constant *e*. As it turns out, *e* can be approximated by the sum of the series $(1/n!)$ where *n* varies from 0 to infinity. A Jack program that uses this method to computes *e* up to a given accuracy (controlled by *n*) is given in Prog. 8-4.

```
/** This program approximates the constant e using the
    formula e=sigma(1/n!) where n goes from 0 to infinity. */

class Main {

  /** Recursive factorial function that returns n! */
  function int factorial(int n){
    if (n=0) {
       return 1;
    }
    else {
       return n*Main.factorial(n-1);
    }
  }

  function void main(){
    var int n;
    var Fraction e;
    let n=0;
    let e=Fraction.new(0,1);  // start with e=0
    // we will use 8 iterations
    while (n<8) {
      let e=Fraction.add(e, Fraction.new(1, Main.factorial(n)));
      let n=n+1;
    }
    do Output.printString("e is approximately ");
    do e.print();
    do Output.println();
    return;
  }

} // end Main
```

PROGRAM 8-4: More examples of Jack programming. Note that statement blocks must be enclosed in brackets even if they include a single statement. Also note that each Jack statement begins with a syntactic prefix like `let` and `do`. These conventions were introduced to the language in order to simplify the writing of Jack compilers.

Since many programmers will likely need to use the factorial function and the *e* constant in many different programs, it would make sense to deliver this functionality through a general library of math-oriented services. Such a library could be implemented as a Jack class, say `Math`, that will implement `function int factorial(int n)` and `function Fraction e(int n)`. This way, Jack programmers could get an approximation of *e* through function calls like `Math.e(12)`, where 12 (an arbitrary choice) determines the desired accuracy. In a similar fashion, suppose we have to compute $17 \cdot (x+5)! - y$ and store the result in variable *z*. This can be done using the Jack statement `let z=17*Math.factorial(x+5)-y`. We now turn to a formal description of the Jack language, focusing on its syntactic elements, program structure, variables, expressions, and statements.

8.2.2 Syntactic Elements

A Jack program is a sequence of tokens. Tokens are separated by an arbitrary amount of white space (including comments), that is ignored. Tokens can be *symbols*, *reserved words*, *constants*, and *identifiers*, as listed in Table 8-5. The language is not case-sensitive.

White space and comments	Space characters, newline characters, and *comment*s are ignored. The following *comment* formats are supported:
	`//  comment to end of line`
	`/*  comment until closing */`
	`/** API documentation comment */`

Symbols	`( )` : arithmetic grouping, and parameter/argument-lists grouping `[ ]` : array indexing; `{ }` : program structure and statement grouping; `,` : variable list separator; `;` : statement terminator; `=` : assignment and comparison operator; `.` : class membership; `+ - * / & | ~ < >` : operators.
Reserved words	`class, constructor, method, function:` Program components `int, boolean, char, void:` Primitive types `var, static, field:` Variable declarations `let, do, if, else, while, return:` Statements `true, false, null:` Constant values `this:` Object reference `Array, String, Sys, Memory, Math, Output, Keyboard, Screen:` Built-in Jack classes
Constants	*Integer* constants are in standard decimal notation, e.g. "`1984`". An expression like "`-13`" is not a constant, but rather a unary minus operator applied to an integer constant. *String* constants are enclosed within two quote (") characters and may contain any characters except *newline* or *double-quote*. These characters can be obtained using the function calls `String.newLine()` and `String.doubleQuote()`. *Boolean* constants can be `true` or `false`. The constant `null` signifies a null reference.
Identifiers	Identifiers are composed from arbitrarily long sequences of letters (`A-Z`, `a-z`), digits (`0-9`), and "_". The first character must be a letter or "_". The language is case insensitive.

TABLE 8-5: Syntactic elements of the Jack language.

8.2.3 Program Structure

The basic programming unit in Jack is a *class*. Each class resides in a separate file and can be compiled separately.

Class declarations have the following format:

```
class name {
    field and static variable declarations  // must  precede the subroutine declarations
    subroutine declarations  // a sequence of constructor, method, and function declarations
}
```

Each class declaration begins with a name through which the class can be globally accessed. Next comes an optional sequence of class-level *fields* and *static variables*, described below. Next comes a sequence of *subroutine* declarations, defining *methods*, *functions*, and *constructors*. A method "belongs" to an object and provides the objects' functionality, while a function "belongs" to the class in general and is not associated with a particular object (similar to Java's *static methods*). A constructor "belongs" to the class and generates object instances of this class.

Subroutine declarations have the following formats:

```
constructor type name (parameter-list) {
    declarations
    statements
}

method type name (parameter-list) {
    declarations
    statements
}

function type name (parameter-list) {
    declarations
    statements
}
```

Each subroutine has a *name* through which it can be accessed. The subroutine's *type* specifies the type of the value returned by the subroutine. As we will see shortly, subroutines can exit via either a void "`return`" statement or a "`return x`" statement where `x` is some value. If the subroutine returns no value, the subroutine type is declared `void`; otherwise it can be any of the primitive or object data types supported by the language (section 8.2.4). Constructors must return the class's own type, i.e. the class name.

Each subroutine contains an arbitrary sequence of variable declarations and then a sequence of statements.

Jack programs: As in Java, a *Jack program* is a collection of one or more classes. One class must be named `Main`, and this class must include at least one function named `main`. When instructed to execute a Jack program that resides in some directory, the host platform will automatically start running the `Main.main` function.

8.2.4 Variables

The variables of object-oriented languages fall into two types: *primitive* variables and *object variables.* Primitive types (e.g. `int`, `char`, `boolean`) are fixed and pre-defined in the language specification. Object types (like `Employee`, `Car`, etc.) are unlimited and are defined by the programmer, as needed.

Variables of primitive types are allocated to memory when they are declared. For example, the declarations "`var int age; var boolean gender;`" will cause the compiler to create the variables `age` and `gender` and to allocate memory space to them. Object variables are treated differently. In Jack (as well as in Java), the declaration of an object variable creates only a reference. Memory may be allocated later, if and when the programmer actually constructs the variable. Example:

```
// The following assumes the existence of Employee and Car classes.
var Employee e;  // creates a reference variable e that contains nothing
var Car c;       // creates a reference variable c that contains nothing
...
let c=Car.new("Jaguar","007")   // creates a new Car(model,license) object
let e=Employee.new("Bond",c)    // creates a new Employee(name,Car) object
// At this point c and e hold the base addresses of the two objects.
```

Each variable has a *name*, a *type* and a *scope*, as we now turn to describe.

Data Types

Primitive types: Jack features three primitive data types:

- `int:` 16-bit 2's complement;
- `boolean:` 0 and -1 signifying *false* and *true* respectively;
- `char:` Unicode character.

Object types: An *object type* is represented by a class name and is implemented as a reference, i.e. an address in memory. For example, consider the declarations "`var Employee e1,e2`" and "`function Fraction e(int n)`". The first example declares two variables `e1` and `e2` that future commands may bind to objects of type `Employee`. The second example declares that the function's return value is an object of type `Fraction`.

Data type conversions: All primitive types are automatically converted to each other as needed. An integer can be assigned to a reference variable, in which case it is treated as an address in memory. The language specification does not specify what happens when a reference is assigned to a variable of an incompatible type; Each Jack implementation may decide whether to allow the conversion or forbid it.

Variable Types

Jack features four types of variables. *Static variables* are defined at the class level, and are shared by all the objects derived from the class. For example, a `BankAccount` class may have a `totalBalance` static variable that holds the sum of all the balances of all the accounts, where each account is represented as an object derived from the `BankAccount` class. *Field variables* are used to define the properties of individual objects derived from the class, e.g. `owner` and `balance` in our banking example. *Local variables*, used by subroutines, exist only as long as the subroutine is running, and *parameters* are used to pass arguments to subroutines. For example, the function signature `function void deposit(BankAccount b,int sum)` indicates that `b` and `sum` are parameters that can be used by the function like local variables. Table 8-6 gives a formal description of all the variable types supported by the Jack language.

Var. Type	Definition and Description	Declared in:	Scope
Static variables	**`static`** ***type name1*****`,`** ***name2*****`, ... ;`** Only one copy of static variables exists, and this copy is shared by all the objects derived from the class. (like *private static variables* in Java)	Class declaration.	The class in which they are declared.
Field variables	**`field`** ***type name1*****`,`** ***name2*****`, ... ;`** Every object of the class has a private copy of the field variables. (like *private variables* in Java)	Class declaration.	The class in which they are declared, but cannot be used in functions.
Local variables	**`var`** ***type name1*****`,`** ***name2*****`, ... ;`** Local variables are allocated on the stack when the subroutine is called and are freed when it returns. (like *method variables* of Java).	Subroutine declaration.	The subroutine in which they are declared.
Parameters	***type name1*****`,`** ***type name2*****`,`** … e.g.: `function void drive (`**`Car c, int miles`**`) { ... }` Used as inputs of subroutines.	Not explicitly declared, but rather appear in parameter-lists of subroutine declarations.	The subroutine in which they are defined.

TABLE 8-6: Variable types in the Jack language.

8.2.5 Expressions

Expression Syntax: Jack expressions are defined recursively according to the rules given in Table 8-7.

A Jack expression is one of the following:

- A *constant*
- A *variable name* in scope (the variable may be *static*, *field*, *local*, or a *parameter*)
- The `this` keyword, denoting the current object. Cannot be used in functions.
- An *array element* using the syntax *name*[*expression*], where *name* is a variable name of type `Array` in scope.
- A *subroutine call* that returns a non-void type (see Section 8.2.6).
- An expression prefixed by one of the unary operators {-,~}:
  -*expression*: arithmetic negation
  ~*expression*: boolean negation (bitwise for integers)
- An expression of the form *expression operator expression* where *operator* is one of the binary operators {+,-,*,/,&,|,<,>,=}:
  + - * / : integer arithmetic operators
  & | : boolean `And`, `Or` (bitwise for integers) operators
  < > = : comparison operators
- An expression in parenthesis: **(***expression***)**

TABLE 8-7: Jack expressions are either atomic or derived recursively from simpler expressions

Order of evaluation and operator priority: Expressions are evaluated left to right, except that expressions in parentheses are evaluated first. Operator priority is ***not*** defined by the language. Thus an expression like `2+3*4` will yield 20, whereas `2+(3*4)` will yield 14. The need to use parentheses in such expressions makes Jack programming a bit cumbersome. At the same time, the lack of operator priority makes the writing of Jack compilers simpler. Of course different Jack implementations can implement any desired operators priority. In other words, we have shifted the issue of operator priority from the language specification level to the language implementation level.

8.2.6 Subroutine calls

Subroutine calls invoke methods, functions, and constructors for their effect, using the general syntax *subroutineName*(*argument-list*). The number and type of the arguments must match those of the subroutine's parameters, as defined in its declaration. The parentheses must appear even if the argument list is empty. Each argument may be an expression of unlimited complexity. For example, consider the function declaration "`function int sqrt(int n)`", designed to return the integer part of the square root of its single parameter. Such a function can be invoked using calls like `sqrt(17)`, `sqrt(x)`, `sqrt(a*c-17)`, `sqrt(a*sqrt(c-17)+3)` and so on.

Within a class, methods are called using the syntax *methodName*(*argument-list*), while functions and constructors must be called using their full-names, i.e. *className.subroutineName*(*argument-list*). Outside a class, the class functions and constructors are also called using the full-name syntax, while methods are called using the syntax *var.methodName(argument-list)*, where *var* is either a previously defined object variable or the `this` reserved word (as in Java). Prog. 8-8 gives some examples.

```
class Bar {
...
}

class Foo {
  ...
  function void f() {
    var Bar b;
    ...
    ... g(5,7)    // call to method g of class Foo (on this object)
    ... Foo.p(2)  // call to function or constructor p of class Foo
    ... Bar.h(3)  // call to function or constructor h of class Bar
    ... b.q()     // call to method q of class Bar (on object b)
    ...
  }
  ...
}
```

PROGRAM 8-8: Subroutine call examples.

Object Construction: A variable of an object type is implemented as a reference to an object of this type. To actually create the object, a class constructor must be called. When the constructor is called, the compiler allocates memory space for the new object, and assigns the allocated space's base address to `this`. Thus a constructor is used just like a function of the class -- a function that happens to return a reference to the newly created object.

Example: Consider a class named `List`, designed to hold an element and a reference to the rest of the list. Prog. 8-9 gives a possible `List` implementation.

```
/** A List class (partial implementation) */
class List {
  field int data;
  field List next;

  constructor List new(int car, List cdr){
    let data = car;
    let next = cdr;
    return this;
  }
...
}

// create a list holding the numbers (2,3,5):
function void create235(){
  Var List v;
  Let v=List.new(5,null);
  Let v=List.new(3,v);
  Let v=List.new(2,v);
  ...
}
```

PROGRAM 8-9: Examples of object construction code in Jack. If we use `()` to denote `null` and `(car,cdr)` to denote a list, then the function constructs the list `(2,(3,(5,())))`.

8.2.7 Statements

The Jack language features 5 statements. They are defined and described in Table 8-10.

Statement	Syntax	Description
let	**let** *variable*=*expression*; or **let** *variable* [*expression*]=*expression*;	An assignment operation. The variable may be static, local, field, or a parameter. Alternatively the variable may be an array entry.

if	**if** (*expression*) {
	statements
	}
	else {
	statements
	}
		Typical "if" statement.
		The `else` clause is optional.
		The brackets are mandatory even if "statements" is a single statement.

while	**while** (*expression*) {
	statements
	}
		Typical "while" statement.
		The brackets are mandatory even if "statements" is a single statement.

do	**do** *function-or-method-call*;	Used to call a function or method for their effect, ignoring the value they return, if any. The function or method call follows the syntax described in section 8.2.6.
return	**return** *expression*; or **return**;	Used to return values from subroutines. The second form must be used by functions and by methods that return a void value. Constructors must return the expression `this`.

TABLE 8-10: Jack statements.

8.3 Jack Language Extensions

The Jack language comes in two versions. The basic version was described so far. The full version includes various language extensions, also called libraries, designed to augment the language's functionality. The libraries are implemented as Jack classes, written in the basic language. The full language implementation assumes the following classes:

- **Math:** Provides basic mathematical operations;
- **String:** Implements the `String` type and basic string-related operations;
- **Array:** Enables the construction and disposal of arrays;
- **Output:** Handles text based output;
- **Screen:** Handles graphic screen output;
- **Keyboard:** Handles user input form the keyboard;
- **Memory:** Handles memory operations;
- **Sys:** Provides some execution-related services.

This section specifies the subroutines that are supposed to be in these classes.

Math

This class enables various mathematical operations.

- Function int **`abs`**(int x): Returns the absolute value of x.
- Function int **`multiply`**(int x, int y): Returns the product of x and y.
- Function int **`divide`**(int x, int y): Returns the integer part of the x/y.
- Function int **`min`**(int x, int y): Returns the minimum of x and y.
- Function int **`max`**(int x, int y): Returns the maximum of x and y.
- Function int **`sqrt`**(int x): Returns the integer part of the square root of x.

String

This class implements the `String` data type and various string-related operations.

- Constructor String **`new`**(int maxLength): Constructs a new empty string (of length zero) that can contain at most maxLength characters.
- Method void **`dispose`**(): Disposes this string.
- Method int **`length`**(): Returns the length of this string.
- Method char **`charAt`**(int j): Returns the character at location j of this string.
- Method void **`setCharAt`**(int j, char c): Sets the j'th element of this string to c.
- Method String **`appendChar`**(char c): Appends c to this string and returns this string.
- Method void **`eraseLastChar`**(): Erases the last character from this string.

- Method int **intValue**(): Returns the integer value of this string (or at least the prefix part, i.e. until a non numeric character is found).
- Method void **setInt**(int j): Sets this string to hold a representation of j.
- Function char **backSpace**(): Returns the backspace character.
- Function char **doubleQuote**(): Returns the double quote (“) character.
- Function char **newLine**(): Returns the newline character.

Array

This class enables the construction and disposal of arrays.

- Function Array **new**(int size): Constructs a new array of the given size.
- Method void **dispose**(): Disposes this array.

Output

This class allows writing text on the screen.

- Function void **moveCursor**(int i, int j): Moves the cursor to the j’th column of the i’th row, and erases the character located there.
- Function void **printChar**(char c): Prints c at the cursor location and advances the cursor one column forward.
- Function void **printString**(String s): Prints s starting at the cursor location, and advances the cursor appropriately.
- Function void **printInt**(int i): Prints i starting at the cursor location, and advances the cursor appropriately.
- Function void **println**(): Advances the cursor to the beginning of the next line.
- Function void **backSpace**(): Moves the cursor one column back.

Screen

This class allows drawing graphics on the screen. It accesses the screen directly, using the screen’s memory-mapped address as defined by the hardware.

- Function void **clearScreen**(): Erases the entire screen.
- Function void **setColor**(boolean b): Sets the screen color (white=false, black=true) to be used for all further drawXXX commands.
- Function void **drawPixel**(int x, int y): Draws the (x,y) pixel.
- Function void **drawLine**(int x1, int y1, int x2, int y2): Draws a line from pixel (x1,y1) to pixel (x2,y2).
- Function void **drawRectangle**(int x1, int y1, int x2, int y2): Draws a filled rectangle where the top left corner is (x1, y1) and the bottom right corner is (x2,y2).

- Function void **drawCircle**(int x, int y, int r): Draws a filled circle of radius r around (x,y)

Keyboard

This class allows reading inputs from the keyboard.

- Function char **keyPressed**(): Returns the character of the currently pressed key on the keyboard; if no key is currently pressed, returns 0.
- Function char **readChar**(): Waits until a key is pressed on the keyboard and released, and then echoes the key to the screen and returns the character of the pressed key.
- Function String **readLine**(String message): Prints the message on the screen, reads the next line (until a newline character) from the keyboard, echoes the line to the screen, and returns its value. This method handles user backspaces.
- Function String **readInt**(String message): Prints the message on the screen, reads the next line (until a newline character) from the keyboard, echoes the line to the screen, and returns the integer until the first non numeric character in the line. This method handles user backspaces.

Memory

This class allows direct access to the main memory.

- Function int **peek**(int address): Returns the value of the main memory at this address.
- Function void **poke**(int address, int value): Sets the value of the main memory at this address to the given value.
- Function Array **alloc**(int size): Allocates the specified space on the heap and returns a reference to it.
- Function void **deAlloc**(Array o): De-allocates the given object and frees its memory space.

Sys

This class supports some execution-related services.

- Function void **halt**(): Halts the program execution.
- Function void **error**(int errorCode): Prints the error code on the screen and halts.
- Function void **wait**(int duration): Waits approximately *duration* milliseconds and returns.

8.4 Applications

Jack is a general-purpose language that can be implemented over different hardware platforms. In the next two chapters we will develop a Jack compiler that ultimately generates Hack code, and thus it is natural to discuss Jack applications in the context of the Hack platform.

Recall that the Hack computer is equipped with a 256 rows by 512 columns screen and a standard keyboard. These I/O devices, along with the Jack classes that drive them, enable the development of interactive programs with a graphical GUI. Figure 8-11 gives some examples.



FIGURE 8-11: Screen shots of three computer games written in Jack, running on the Hack computer. Left to right: a single player "Pong" game, after scoring three points (the authors's record), a "Hangman" game, where the user has to guess the name of the capital city hidden behind the squares (the 1-9 number determines the game level), and a maze game, in which the user scores points by moving the square over the dots.

The Pong game: A ball is moving on the screen randomly, bouncing off the screen "walls". The user can move a small bat horizontally by pressing the left and right arrow keys. Each time the bat hits the ball, the user scores a point and the bat shrinks a little, to make the game harder. If the user misses and the ball hits the bottom horizontal line, the game is over.

The Pong game provides a good illustration of Jack programming over the Hack platform. First, it is a non-trivial program, requiring several hundreds lines of Jack code organized in several classes. Second, the program has to carry out some non-trivial mathematical calculations, in order to compute the direction of the ball's movements. Third, the program must animate the movement of graphical objects on the screen (the bat and the ball), requiring extensive use of the language's graphics drawing methods. Finally, in order to do all of the above *quickly*, the program must be efficient, meaning that it has to do as few real-time calculations and screen drawing operations as possible.

Other applications: Of course Pong is just one example of the numerous applications that can be written in Jack over the Hack platform. Since the Jack screen resembles that of a cellular telephone, it lends itself nicely to the computer games that one normally finds on cellular telephones, e.g. *Snake* and *Tetris*. In general, the more event- and GUI-driven is the application, the more "catchy" it will be.

Having said that, recall that the Jack/Hack platform is in fact a general-purpose computer. As such, one can use it to develop any application that involves inputs, outputs, and calculations, not necessarily in this order. For example, one can write a program that inputs student names and grades, and then prints the average grade and the name of the student who got the maximal grade, and so on. We now turn to describe how such applications can be planned and developed.

8.5 Specification and Implementation

The development of Jack applications over a target platform requires careful planning (as always). In the specification stage, the application designer must consider the physical limitations of the target hardware, and plan accordingly. For example, the physical dimensions of the computer's screen limits the size of the graphical images that the program can handle. Likewise, one must consider the language's range of input/output commands, in order to have a realistic appreciation of what can and cannot be done.

Although the language's capabilities can be extended at will (by modifying its supporting libraries, also written in Jack), readers will perhaps want to hone their programming skills elsewhere. After all, we don't expect Jack to be part of your life beyond this course. Therefore, it is best to view the Jack/Hack platform as a given environment, and make the best out of it. That's precisely what programmers do when they write software for embedded devices and dedicated processors that operate in restricted environments. Instead of viewing the constrains imposed by the host platform as a problem, they view it an opportunity to display their resourcefulness and ingenuity. That's why some of the best programmers in the trade were first trained on primitive computers.

Specification: The specification stage should start with some description of the application's behavior and, in the case of graphical and interactive applications, some hand-written drawings of typical screens. Using object-oriented analysis and design practices, the designer should then create an object-based architecture of the application. This requires the identification of *classes*, *fields*, and *subroutines*, resulting with a well-defined API (like Prog. 8-1).

Implementation: Next, one implements the API in Jack and tests it on the target platform. Following compilation into VM code, the program can be tested in two alternatives ways. Either the VM code can be executed directly on a VM emulator, or it can be translated into Hack code and ran on the hardware platform.

8.6 Perspective

Work in progress.

8.7 Build It

Unlike all the other projects in the book, this project does not involve building part of the computer's hardware or software systems. Rather, you have to choose an application of your choice, specify it, and then implement it in Jack on the Hack platform.

Objective: The major objective of this project is to get acquainted with the Jack language, for two purposes. First, you have to know Jack intimately in order to write the Jack compiler in Projects 9 and 10. Second, you have to be familiar with Jack's supporting libraries in order to write the computer's operating system in Project 11. The best way to gain this knowledge is to write a Jack application.

Contract: Adopt or invent an application idea, e.g. a simple computer game or some other interactive program. Then specify and build the application.

Steps

1. Download the `os.zip` file and extract its contents to a directory named `project8` on your computer (you may want to give this directory a more descriptive name, e.g. the name of your program). The resulting set of `.vm` files constitute an implementation of the computer's operating system, including all the supporting Jack libraries.
2. Write your Jack program (set of one or more Jack classes) using a plain text editor. Put each class in a separate `ClassName.jack` file. Put all the `.jack` files (of the same program) in the same program directory described in step 1.
3. Compile your program using the supplied Jack Compiler. This is best done by applying the compiler to the name of the program directory. This will cause the compiler to translate all the `.jack` classes in that directory to corresponding `.vm` files. If a compilation error is reported, debug the program and re-compile until no error messages are issued.
4. At this point, the program directory should contain three sets of files: (a) your source `.jack` files, (b) a compiled `.vm` file for each one of your source `.jack` files, and (c) the supplied operating system `.vm` files. To test the compiled program, invoke the *VM Emulator* and direct it to load the program by selecting the program directory name. Then execute the program. In case of run-time errors or undesired program behavior, fix the program and go back to stage 3.

Deliverables

Different courses may require different deliverables. At minimum, you should deliver a `readme.txt` text file that tells users everything they have to know about using your program, and a zip file containing all your source Jack files. Do not submit any `.vm` files.